```
> with(OreModules):
> with(Stafford):
```

We consider the ordinary differential linear system which defines the derivative of the Dirac distribution δ. For more details, see Example 4 of A. Quadrat and D. Robertz, *Constructive computation of bases of free modules over the Weyl algebra, INRIA Report 5786,* 2005. We first introduce the first Weyl algebra $Alg = A\_1$ formed by ordinary differential operators with polynomial coefficients.

```
> Alg:=DefineOreAlgebra(diff=[D,t],polynom=[t]):
```

The system is then defined by the following matrix of differential operators

```
> R:=evalm([[t^2],[t*D+2]]);
```

$$R := \begin{bmatrix} t^2 \\ t\,\mathrm{D} + 2 \end{bmatrix}$$

i.e., the derivative y of the Dirac distribution δ satisfies the equations:

```
> ApplyMatrix(R,[y(t)],Alg)=evalm([[0]$2]);
```

$$\begin{bmatrix} t^2\,\mathrm{y}(t) \\ 2\,\mathrm{y}(t) + t\,(\frac{d}{dt}\,\mathrm{y}(t)) \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

Let us compute a finite free resolution of the left Alg-module M defined as the cokernel of R, i.e., $M = Alg/(\mathrm{Alg}\ (t^2) + \mathrm{Alg}\ (t\,\mathrm{D}+2))$:

```
> F:=FreeResolution(R,Alg);
```

$$F := \mathrm{table}([1 = \begin{bmatrix} t^2 \\ t\,\mathrm{D} + 2 \end{bmatrix}, 2 = \begin{bmatrix} \mathrm{D} & -t \end{bmatrix}, 3 = \mathrm{INJ}(1)])$$

Let us check whether or not we can find a shorter free resolution of M:

```
> G:=ShorterFreeResolution(F,Alg);
```

$$G := \mathrm{table}([1 = \begin{bmatrix} t^2 & t \\ t\,\mathrm{D} + 2 & \mathrm{D} \end{bmatrix}, 2 = \mathrm{INJ}(2)])$$

As the first matrix of G has a trivial syzygy module, we obtain that G is a "minimal free resolution" of M. This last result can be directly obtained by:

```
> MinimalFreeResolution(R,Alg);
```

$$\mathrm{table}([1 = \begin{bmatrix} t^2 & t \\ t\,\mathrm{D} + 2 & \mathrm{D} \end{bmatrix}, 2 = \mathrm{INJ}(2)])$$

Now, we can check whether or not M is a stably free left Alg-module by checking if a right-inverse of $G[1]$ exists:

```
> RightInverse(G[1],Alg);
```

$$[]$$

Hence, we obtain that M is not a stably free left Alg-module. Another way to check this result is to compute the projective dimension of M.

```
> ProjectiveDimension(R,Alg);
```

$$1$$

We obtain again that M is a not a stably free left *Alg*-module. Finally, this result can also be checked by computing:

```
> Ext1:=Exti(Involution(R,Alg),Alg,1);
```

$$Ext1 := [\left[\begin{array}{c} t^2 \\ t\,\mathrm{D}+2 \end{array}\right], [\ 1\ ], \mathrm{SURJ}(1)]$$

As the first matrix of *Ext1* is not the identity matrix, we conclude that M is not a torsion-free *Alg*-module, and thus, not a stably free *Alg*-module. As the second matrix is just 1, we obtain that M is a torsion *Alg*-module.

Finally, we can prove that the left ideal of *Alg* defined by the two entries of the matrix R is not principal. For more details, see Example 17 of A. Quadrat and D. Robertz, *Constructive computation of bases of free modules over the Weyl algebra; INRIA Report 5786,* 2005.

We now consider the linear system of partial differential equations formed by the infinitesimal transformations of the Lie pseudogroup defining the contact transformations. See Example V. 1. 84 of J.-F. Pommaret, *Partial Differential Control Theory,* Kluwer, 2001, and Example 5 of A. Quadrat and D. Robertz, *Constructive computation of bases of free modules over the Weyl algebra, INRIA Report 5786,* 2005.

We first introduce the Weyl algebra $Alg2 = A\_3$ of the partial differential operators in $x1$, $x2$ and $x3$ with polynomial coefficients.

```
> Alg2:=DefineOreAlgebra(diff=[D1,x1],diff=[D2,x2],diff=[D3,x3],polynom=[x1,x2,x3]):
```

The system is defined by the following matrix *R2* of differential operators:

```
> R2:=evalm([[(x2/2)*D1,x2*D2+1,x2*D3+D1/2],[-(x2/2)*D2-3/2,0,D2/2],
> [-D1-(x2/2)*D3,-D2,-D3/2]]);
```

$$R2 := \left[\begin{array}{ccc} \frac{x2\,\mathrm{D1}}{2} & x2\,\mathrm{D2}+1 & x2\,\mathrm{D3}+\frac{\mathrm{D1}}{2} \\ -\frac{x2\,\mathrm{D2}}{2}-\frac{3}{2} & 0 & \frac{\mathrm{D2}}{2} \\ -\mathrm{D1}-\frac{x2\,\mathrm{D3}}{2} & -\mathrm{D2} & -\frac{\mathrm{D3}}{2} \end{array}\right]$$

In other words, we consider the system of partial differential equations defined by:

```
> x:=x1,x2,x3:
> ApplyMatrix(R2,[eta[1](x),eta[2](x),eta[3](x)],Alg2)=evalm([[0]$3]);
```

$$\begin{array}{l} \left[\frac{1}{2}\,x2\,(\frac{\partial}{\partial x1}\,\eta_1(x1,\,x2,\,x3))+\eta_2(x1,\,x2,\,x3)+x2\,(\frac{\partial}{\partial x2}\,\eta_2(x1,\,x2,\,x3)) \right. \\ \left. +\,\frac{1}{2}\,(\frac{\partial}{\partial x1}\,\eta_3(x1,\,x2,\,x3))+x2\,(\frac{\partial}{\partial x3}\,\eta_3(x1,\,x2,\,x3))\right] \\ \left[-\frac{3}{2}\,\eta_1(x1,\,x2,\,x3)-\frac{1}{2}\,x2\,(\frac{\partial}{\partial x2}\,\eta_1(x1,\,x2,\,x3))+\frac{1}{2}\,(\frac{\partial}{\partial x2}\,\eta_3(x1,\,x2,\,x3))\right] \\ \left[-\,(\frac{\partial}{\partial x1}\,\eta_1(x1,\,x2,\,x3))-\frac{1}{2}\,x2\,(\frac{\partial}{\partial x3}\,\eta_1(x1,\,x2,\,x3))-(\frac{\partial}{\partial x2}\,\eta_2(x1,\,x2,\,x3)) \right. \\ \left. -\,\frac{1}{2}\,(\frac{\partial}{\partial x3}\,\eta_3(x1,\,x2,\,x3))\right] = \left[\begin{array}{c} 0 \\ 0 \\ 0 \end{array}\right] \end{array}$$

Let us compute a finite free resolution of the left *Alg2*-module *M2* defined as the cokernel of the matrix *R2*, i.e., *M2=Alg2^{1*3}/(Alg2^{1*3} R_2)*.

```
> F2:=FreeResolution(R2,Alg2);
```

$$F2 := \mathrm{table}([1 = \begin{bmatrix} \frac{x2\,\mathrm{D1}}{2} & x2\,\mathrm{D2}+1 & x2\,\mathrm{D3}+\frac{\mathrm{D1}}{2} \\ -\frac{x2\,\mathrm{D2}}{2}-\frac{3}{2} & 0 & \frac{\mathrm{D2}}{2} \\ -\mathrm{D1}-\frac{x2\,\mathrm{D3}}{2} & -\mathrm{D2} & -\frac{\mathrm{D3}}{2} \end{bmatrix},$$

$$2 = \begin{bmatrix} \mathrm{D2} & -\mathrm{D1}-x2\,\mathrm{D3} & 2+x2\,\mathrm{D2} \end{bmatrix},$$

$$3 = \mathrm{INJ}(1)$$

$$])$$

Let us check whether or not *M2* admits a shorter finite free resolution. We have:

```
> G2:=ShorterFreeResolution(F2,Alg2);
```

$$G2 := \mathrm{table}([1 = \begin{bmatrix} \frac{x2\,\mathrm{D1}}{2} & x2\,\mathrm{D2}+1 & x2\,\mathrm{D3}+\frac{\mathrm{D1}}{2} & -x2 \\ -\frac{x2\,\mathrm{D2}}{2}-\frac{3}{2} & 0 & \frac{\mathrm{D2}}{2} & 0 \\ -\mathrm{D1}-\frac{x2\,\mathrm{D3}}{2} & -\mathrm{D2} & -\frac{\mathrm{D3}}{2} & 1 \end{bmatrix}, 2 = \mathrm{INJ}(3)])$$

Hence, we obtain a shorter finite free resolution defined by *G2*. As the first matrix G2 has full row rank, i.e., its syzygy module is trivial as *G2*[2]= INJ(3), we know that we cannot reduce once more the free resolution *G2*. This information can also be obtained using the command "minimal free resolution".

```
> MinimalFreeResolution(R2,Alg2);
```

$$\mathrm{table}([1 = \begin{bmatrix} \frac{x2\,\mathrm{D1}}{2} & x2\,\mathrm{D2}+1 & x2\,\mathrm{D3}+\frac{\mathrm{D1}}{2} & -x2 \\ -\frac{x2\,\mathrm{D2}}{2}-\frac{3}{2} & 0 & \frac{\mathrm{D2}}{2} & 0 \\ -\mathrm{D1}-\frac{x2\,\mathrm{D3}}{2} & -\mathrm{D2} & -\frac{\mathrm{D3}}{2} & 1 \end{bmatrix}, 2 = \mathrm{INJ}(3)])$$

Finally, let us check whether or not *M2* is a projective, and thus, a stably free left *Alg2*-module. In order to do that, let us compute if the full row rank matrix *G2*[1] admits a right-inverse:

```
> RightInverse(G2[1],Alg2);
```

$$\begin{bmatrix} 0 & -1 & 0 \\ 1 & 0 & x2 \\ 0 & -x2 & 0 \\ \mathrm{D2} & -\mathrm{D1}-x2\,\mathrm{D3} & 2+x2\,\mathrm{D2} \end{bmatrix}$$

This last result is also coherent with the fact that the projective dimension of *M2* is:

```
> ProjectiveDimension(R2,Alg2);
```

$$0$$

Indeed, *M2* is then a projective left *Alg2*-module. We note that this result corrects a small typo in Example 7 in page 14 of A. Quadrat and D. Robertz, *Constructive computation of bases of free modules overthe Weyl algebra*, *INRIA Report 5786,* 2005. Hence, *M2* is a stably free *Alg2*-module of rank:

```
> OreRank(R2,Alg2);
```

$$1$$

We cannot use the result due to J. T. Stafford which asserts that every stably free *Alg2*-module of rank at least 2 is free, in order to conclude that *M2* is a free left *Alg2*-module. However, we can try to find if there exists an injective minimal parametrization of *R2*:

```
> Q:=MinimalParametrization(R2,Alg2);
```

$$Q := \begin{bmatrix} -\mathrm{D2} \\ \mathrm{D1} + x2\,\mathrm{D3} \\ -2 - x2\,\mathrm{D2} \end{bmatrix}$$

```
> S:=LeftInverse(Q,Alg2);
```

$$S := \begin{bmatrix} \frac{x2}{2} & 0 & \frac{-1}{2} \end{bmatrix}$$

```
> Mult(S,Q,Alg2);
```

$$\begin{bmatrix} 1 \end{bmatrix}$$

Hence, we obtain that *M2* is a free *Alg2*-module of rank 1 and we have the following parametrization of the system $R2\,\eta = 0$:

```
> evalm([seq([eta[i](x)],i=1..3)])=Parametrization(R2,Alg2);
```

$$\begin{bmatrix} \eta_1(x1,\, x2,\, x3) \\ \eta_2(x1,\, x2,\, x3) \\ \eta_3(x1,\, x2,\, x3) \end{bmatrix} = \begin{bmatrix} -(\frac{\partial}{\partial x2}\,\xi_1(x1,\, x2,\, x3)) \\ (\frac{\partial}{\partial x1}\,\xi_1(x1,\, x2,\, x3)) + x2\,(\frac{\partial}{\partial x3}\,\xi_1(x1,\, x2,\, x3)) \\ -2\,\xi_1(x1,\, x2,\, x3) - x2\,(\frac{\partial}{\partial x2}\,\xi_1(x1,\, x2,\, x3)) \end{bmatrix}$$

This parametrization is injective as we then have:

```
> xi[1](x)=ApplyMatrix(S,[seq(eta[i](x),i=1..3)],Alg2)[1,1];
```

$$\xi_1(x1,\, x2,\, x3) = \frac{1}{2}\,x2\,\eta_1(x1,\, x2,\, x3) - \frac{1}{2}\,\eta_3(x1,\, x2,\, x3)$$

Let us consider the so-called *Janet's example* (M. Janet, *Leçons sur les systemes d'equations aux derivees partielles*, Gauthier-Vilars, 1929, p. 76-77). The system is defined by the following matrix of differential operators:

```
> R3:=evalm([[D3^2-x2*D1^2],[D2^2]]);
```

$$R3 := \begin{bmatrix} \mathrm{D3}^2 - \mathrm{D1}^2\,x2 \\ \mathrm{D2}^2 \end{bmatrix}$$

Let us compute a finite free resolution of the left *Alg2*-module *M3* defined as the cokernel of the matrix *R3*, i.e., *M3*= *Alg2*/(*Alg2* $(\mathrm{D3}^2 - x2\,\mathrm{D1}^2)$+ *Alg2* $(\mathrm{D2}^2)$).

```
> F3:=FreeResolution(R3,Alg2);
```

$$F3 := \mathrm{table}([1 = \begin{bmatrix} \mathrm{D3}^2 - \mathrm{D1}^2\, x2 \\ \mathrm{D2}^2 \end{bmatrix},$$
$$2 = [\mathrm{D2}^3,\ 3\,\mathrm{D1}^2 + \mathrm{D2}\,\mathrm{D1}^2\, x2 - \mathrm{D2}\,\mathrm{D3}^2]$$
$$[\mathrm{D3}^4\,\mathrm{D2}^2 + 2\,\mathrm{D2}\,\mathrm{D3}^2\,\mathrm{D1}^2 - 2\,\mathrm{D3}^2\,\mathrm{D1}^2\, x2\,\mathrm{D2}^2 + 2\,\mathrm{D1}^4 - 2\,\mathrm{D1}^4\, x2\,\mathrm{D2}$$
$$+ \mathrm{D1}^4\, x2^2\,\mathrm{D2}^2,\ 3\,\mathrm{D3}^4\,\mathrm{D1}^2\, x2 - \mathrm{D3}^6 - 3\,\mathrm{D3}^2\,\mathrm{D1}^4\, x2^2 + \mathrm{D1}^6\, x2^3],$$
$$3 = \begin{bmatrix} \mathrm{D1}^4\, x2^2 - 2\,\mathrm{D1}^2\, x2\,\mathrm{D3}^2 + \mathrm{D3}^4 & -\mathrm{D2} \end{bmatrix},$$
$$4 = \mathrm{INJ}(1)$$
$$])$$

We now can compute a shorter free resolution of *M3* if it exists. We obtain:

```
> MinimalFreeResolution(R3,Alg2);
```

$$\mathrm{table}([1 = \begin{bmatrix} \mathrm{D3}^2 - \mathrm{D1}^2\, x2 \\ \mathrm{D2}^2 \end{bmatrix},$$
$$2 = [\mathrm{D2}^3,\ 3\,\mathrm{D1}^2 + \mathrm{D2}\,\mathrm{D1}^2\, x2 - \mathrm{D2}\,\mathrm{D3}^2]$$
$$[\mathrm{D3}^4\,\mathrm{D2}^2 + 2\,\mathrm{D2}\,\mathrm{D3}^2\,\mathrm{D1}^2 - 2\,\mathrm{D3}^2\,\mathrm{D1}^2\, x2\,\mathrm{D2}^2 + 2\,\mathrm{D1}^4 - 2\,\mathrm{D1}^4\, x2\,\mathrm{D2}$$
$$+ \mathrm{D1}^4\, x2^2\,\mathrm{D2}^2,\ 3\,\mathrm{D3}^4\,\mathrm{D1}^2\, x2 - \mathrm{D3}^6 - 3\,\mathrm{D3}^2\,\mathrm{D1}^4\, x2^2 + \mathrm{D1}^6\, x2^3],$$
$$3 = \begin{bmatrix} \mathrm{D1}^4\, x2^2 - 2\,\mathrm{D1}^2\, x2\,\mathrm{D3}^2 + \mathrm{D3}^4 & -\mathrm{D2} \end{bmatrix},$$
$$4 = \mathrm{INJ}(1)$$
$$])$$

Hence, it is not possible to shorten the finite free resolution of *M3* defined by *F3*. Therefore, we obtain that the projective dimension of *M3* is:

```
> ProjectiveDimension(R3,Alg2);
```

$$3$$

Finally, let us consider the following example.

```
> Alg4:=DefineOreAlgebra(diff=[D1,x1],diff=[D2,x2],polynom=[x1,x2]):
```

The system is defined by means of the following matrix of differential operators:

```
> R4:=evalm([[D2^3+x2],[D1^2+D2]]);
```

$$R4 := \begin{bmatrix} \mathrm{D2}^3 + x2 \\ \mathrm{D1}^2 + \mathrm{D2} \end{bmatrix}$$

We then have the system of partial differential equations:

```
> x:=x1,x2:  ApplyMatrix(R4,[y(x)],Alg4)=evalm([[0]$2]);
```

$$\begin{bmatrix} x2\,\mathrm{y}(x1,\, x2) + (\frac{\partial^3}{\partial x2^3}\,\mathrm{y}(x1,\, x2)) \\ (\frac{\partial}{\partial x2}\,\mathrm{y}(x1,\, x2)) + (\frac{\partial^2}{\partial x1^2}\,\mathrm{y}(x1,\, x2)) \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

We compute a finite free resolution of the left *Alg4*-module *M4* defined as the cokernel of the matrix *R4*, i.e., *M4*= *Alg4*/(*Alg4* $(\mathrm{D2}^3 + x2)$ + *Alg4* $(\mathrm{D1}^2 + \mathrm{D2})$).

```
> F4:=FreeResolution(R4,Alg4);
```

$$F4 := \mathrm{table}([1 = \left[\begin{array}{c} \mathrm{D2}^3 + x2 \\ \mathrm{D1}^2 + \mathrm{D2} \end{array}\right],$$

$$2 = \left[\begin{array}{cc} -\mathrm{D1}^4 - 2\,\mathrm{D2}\,\mathrm{D1}^2 - \mathrm{D2}^2 & 2 + \mathrm{D1}^2\,\mathrm{D2}^3 + \mathrm{D1}^2\,x2 + \mathrm{D2}^4 + x2\,\mathrm{D2} \\ 1 - \mathrm{D1}^2\,\mathrm{D2}^3 - \mathrm{D2}^4 - \mathrm{D1}^2\,x2 - x2\,\mathrm{D2} & \mathrm{D2}^6 + 2\,x2\,\mathrm{D2}^3 + 3\,\mathrm{D2}^2 + x2^2 \end{array}\right],$$

$$3 = \left[\begin{array}{cc} \mathrm{D2}^3 + x2 & -\mathrm{D1}^2 - \mathrm{D2} \end{array}\right],$$

$$4 = \mathrm{INJ}(1)$$

$$])$$

Let us check whether or not we can shorten the previous finite free resolution of *M4*.

```
> G4:=ShorterFreeResolution(F4,Alg4);
```

$$G4 := \mathrm{table}([1 = \left[\begin{array}{c} \mathrm{D2}^3 + x2 \\ \mathrm{D1}^2 + \mathrm{D2} \\ 0 \end{array}\right],$$

$$2 = \left[\begin{array}{c} -\mathrm{D1}^4 - 2\,\mathrm{D2}\,\mathrm{D1}^2 - \mathrm{D2}^2\,,\ 2 + \mathrm{D1}^2\,\mathrm{D2}^3 + \mathrm{D1}^2\,x2 + \mathrm{D2}^4 + x2\,\mathrm{D2}\,,\ -\mathrm{D1}^2 - \mathrm{D2} \\ 1 - \mathrm{D1}^2\,\mathrm{D2}^3 - \mathrm{D2}^4 - \mathrm{D1}^2\,x2 - x2\,\mathrm{D2}\,,\ \mathrm{D2}^6 + 2\,x2\,\mathrm{D2}^3 + 3\,\mathrm{D2}^2 + x2^2\,,\ -\mathrm{D2}^3 - x2 \end{array}\right],$$

$$3 = \mathrm{INJ}(2)$$

$$])$$

Hence, we can reduce the length of the free resolution *F4* of *M4* by one. Let us try to continue.

```
> H4:=ShorterFreeResolution(G4,Alg4);
```

$$H4 := \mathrm{table}([1 = \left[\begin{array}{ccc} \mathrm{D2}^3 + x2 & 0 & 1 \\ \mathrm{D1}^2 + \mathrm{D2} & 1 & 0 \\ 0 & \mathrm{D2}^3 + x2 & -\mathrm{D1}^2 - \mathrm{D2} \end{array}\right],\ 2 = \mathrm{INJ}(3)])$$

We obtain a finite free resolution of *M4* defined by a full row rank matrix as *H4*[2]= *INJ*(3). Hence, it is a "minimal free resolution" of *M4*. This last result can be directly checked by computing a minimal free resolution of *M4*:

```
> MinimalFreeResolution(R4,Alg4);
```

$$\mathrm{table}([1 = \left[\begin{array}{ccc} \mathrm{D2}^3 + x2 & 0 & 1 \\ \mathrm{D1}^2 + \mathrm{D2} & 1 & 0 \\ 0 & \mathrm{D2}^3 + x2 & -\mathrm{D1}^2 - \mathrm{D2} \end{array}\right],\ 2 = \mathrm{INJ}(3)])$$

As *M4* is presented by means of a full row rank matrix, we then know that *M4* is a stably free left *Alg4*-module iff the presentation matrix *H4*[1] admits a right-inverse. Let us check this last point:

```
> RightInverse(H4[1],Alg4);
```

$$\left[\begin{array}{c} \mathrm{D1}^2 + \mathrm{D2}\,,\ -\mathrm{D2}^3 - x2\,,\ 1 \\ -\mathrm{D1}^4 - 2\,\mathrm{D2}\,\mathrm{D1}^2 - \mathrm{D2}^2\,,\ 2 + \mathrm{D1}^2\,\mathrm{D2}^3 + \mathrm{D1}^2\,x2 + \mathrm{D2}^4 + x2\,\mathrm{D2}\,,\ -\mathrm{D1}^2 - \mathrm{D2} \\ 1 - \mathrm{D1}^2\,\mathrm{D2}^3 - \mathrm{D2}^4 - \mathrm{D1}^2\,x2 - x2\,\mathrm{D2}\,,\ \mathrm{D2}^6 + 2\,x2\,\mathrm{D2}^3 + 3\,\mathrm{D2}^2 + x2^2\,,\ -\mathrm{D2}^3 - x2 \end{array}\right]$$

Therefore, we obtain that *M4* is a stably free left *Alg4*-module of rank:

```
> OreRank(R4,Alg4);
```

$$0$$

However, as the rank of $M4$ over $Alg4$ is 0, we know that $M4$ is also a torsion left $Alg4$-module, and thus, we obtain that $M4$ must be the zero module. The fact that $M4$ is the zero module is equivalent to the existence of a left-inverse $S4$ of $R4$ over $Alg4$ as we can easily check:

```
> S4:=LeftInverse(R4,Alg4);
```

$$S4 := \left[\begin{array}{cc} \mathrm{D1}^2 + \mathrm{D2} & -\mathrm{D2}^3 - x2 \end{array} \right]$$

Indeed, we have $S4\ R4 = 1$:

```
> Mult(S4,R4,Alg4);
```

$$\left[\begin{array}{c} 1 \end{array} \right]$$

Finally, the projective dimension of $M4 = 0$ is:

```
> ProjectiveDimension(R4,Alg4);
```

$$0$$